


WILEY



Algorithm AS 261: Quantiles of The Distribution of The Square of The Sample Multiple-Correlation Coefficient

Author(s): Cherng G. Ding and Rolf E. Bargmann

Source: *Journal of the Royal Statistical Society. Series C (Applied Statistics)*, Vol. 40, No. 1 (1991), pp. 199-202

Published by: [Wiley](#) for the [Royal Statistical Society](#)

Stable URL: <http://www.jstor.org/stable/2347933>

Accessed: 28/04/2014 15:26

Your use of the JSTOR archive indicates your acceptance of the Terms & Conditions of Use, available at <http://www.jstor.org/page/info/about/policies/terms.jsp>

JSTOR is a not-for-profit service that helps scholars, researchers, and students discover, use, and build upon a wide range of content in a trusted digital archive. We use information technology and tools to increase productivity and facilitate new forms of scholarship. For more information about JSTOR, please contact support@jstor.org.



Wiley and Royal Statistical Society are collaborating with JSTOR to digitize, preserve and extend access to *Journal of the Royal Statistical Society. Series C (Applied Statistics)*.

<http://www.jstor.org>

Algorithm AS 261**Quantiles of the Distribution of the Square of the Sample Multiple-correlation Coefficient**

By Cherng G. Ding†

National Chiao Tung University, Taipei, Republic of China

and Rolf E. Bargmann

University of Georgia, Athens, USA

[Received April 1989]

Keywords: Illinois method; Multiple-correlation coefficient; Secant method**Language**

Fortran 77

Description and Purpose

Let X_1, \dots, X_p be distributed as $N_p(\mu, \Sigma)$ and R be the sample multiple-correlation coefficient between X_i and the other $p - 1$ random variables based on a sample of size N . The cumulative distribution function (CDF) of R^2 is

$$M(R^2; p, N, \rho^2) = \int_0^{R^2} g(t) dt, \quad 0 \leq R^2 \leq 1, \quad (1)$$

where ρ denotes the population multiple-correlation coefficient and $g(R^2)$ is the density of R^2 (see Anderson (1984) for its expression). For given values of m ($0 \leq m \leq 1$), p (> 1), N ($> p$) and ρ^2 ($0 \leq \rho^2 \leq 1$), the function subprogram SQMCQ returns the value of R^2 such that $M(R^2; p, N, \rho^2) = m$.

Numerical Method

Let $X = R^2$ and $f(x) = M(x; p, N, \rho^2) - m$. The equation $f(x) = 0$ is to be solved. f is a strictly increasing function and the solution is unique. A modification of the secant method, called the Illinois method (see Dowell and Jarratt (1971) and Kennedy and Gentle (1980)), is used to find the root.

Given two values x_i and x_{i-1} , the next approximation x_{i+1} to the root is determined by

$$x_{i+1} = x_i - \frac{f(x_i)(x_i - x_{i-1})}{\{f(x_i) - f(x_{i-1})\}}. \quad (2)$$

The end points unity and zero serve as two starting values x_0 and x_1 . Iterations are performed according to the following rules.

†*Address for correspondence:* Institute of Management Science, National Chiao Tung University, 4F, 114 Chung-Hsiao W. Road, Section 1, Taipei, Taiwan, Republic of China.

- (a) If $f(x_{i+1})f(x_i) < 0$, then $(x_{i-1}, f(x_{i-1}))$ is replaced by $(x_i, f(x_i))$.
 (b) If $f(x_{i+1})f(x_i) > 0$, then $(x_{i-1}, f(x_{i-1}))$ is replaced by $(x_{i-1}, f(x_{i-1})/2)$.

After these two rules have been applied, $(x_{i+1}, f(x_{i+1}))$ replaces $(x_i, f(x_i))$. The convergence criterion is based on $|x_{i+1} - x_i|$ and $|f(x_{i+1})|$ (relatively).

Structure

REAL FUNCTION SQMCQ(CDF, IP, N, RHO2, IFAULT)

Formal parameters

<i>CDF</i>	Real	input:	the cumulative probability m (at which the quantile is desired)
<i>IP</i>	Integer	input:	the number of random variables p
<i>N</i>	Integer	input:	the sample size N
<i>RHO2</i>	Real	input:	the square of the population multiple-correlation coefficient ρ^2
<i>IFault</i>	Integer	output:	a fault indicator: = 1 if there is no convergence after n iterations in SQMCOR; = 2 if $p < 1$, $p > N$, $\rho^2 < 0$ or $\rho^2 > 1$; = 3 if $m < 0$ or $m > 1$; = 0 otherwise

Auxiliary Algorithms

The auxiliary routine SQMCOR (Ding and Bargmann, 1991) is invoked by function SQMCQ to evaluate the CDF of R^2 . SQMCOR calls algorithm CACM 291 (Pike and Hill, 1966) and algorithm AS 63 (Majumder and Bhattacharjee, 1973).

Constants

The variable EPS in the DATA statement represents a small real number to indicate the convergence criterion. The value given here is 1.0×10^{-6} .

Precision

Double-precision operation may be performed by changing REAL to DOUBLE PRECISION in the REAL statement. The real constants in the DATA statements also need to be in double precision, and the value of EPS may be changed from 1.0×10^{-6} to 1.0×10^{-12} to increase the accuracy. The auxiliary routines must also be converted to double precision.

Time

No absolute timings are given here since the execution time depends entirely on the values of the input parameters.

Acknowledgements

The authors thank the referee and the Algorithms Editor for their valuable comments that led to substantial improvements in the manuscript.

References

- Anderson, T. W. (1984) *An Introduction to Multivariate Statistical Analysis*, 2nd edn, p. 145. New York: Wiley.
- Ding, C. G. and Bargmann, R. E. (1991) Algorithm AS 260: Evaluation of the distribution of the square of the sample multiple-correlation coefficient. *Appl. Statist.*, **40**, 195-198.
- Dowell, M. and Jarratt, P. (1971) A modified *regular falsi* method for computing the root of an equation. *BIT*, **11**, 168-174.
- Kennedy, W. J. and Gentle, J. E. (1980) *Statistical Computing*, p. 74. New York: Dekker.
- Majumder, K. L. and Bhattacharjee, G. P. (1973) Algorithm AS 63: The incomplete beta integral. *Appl. Statist.*, **22**, 409-411.
- Pike, M. C. and Hill, I. D. (1966) Algorithm 291: Logarithm of gamma function. *Commun. Ass. Comput. Mach.*, **9**, 684.

```

REAL FUNCTION SQMCQ(CDF, IP, N, RHO2, IFAULT)
C
C      ALGORITHM AS 261  APPL. STATIST. (1991) VOL. 40, NO. 1
C
C      Returns the quantile of the distribution of the square
C      of the sample multiple correlation coefficient for
C      given values of CDF, IP, N, and RHO2
C
C      A modification of the secant method, called the
C      Illinois method, is used
C
C      The auxiliary algorithm SQMCOR, used to compute the
C      C.D.F. of the square of the sample multiple correlation
C      coefficient, is required
C
C
C      INTEGER IP, N, IFAULT
C      REAL CDF, RHO2
C      REAL DIFF, EPS, F0, F1, F2, X0, X1, X2, ZERO, ONE, TWO
C      REAL SQMCOR
C      EXTERNAL SQMCOR
C      DATA ZERO, ONE, TWO / 0.0, 1.0, 2.0 /
C      DATA EPS / 1.0E-6 /
C
C      SQMCQ = CDF
C      IFAULT = 2
C
C      Perform domain check
C
C      IF (RHO2 .LT. ZERO .OR. RHO2 .GT. ONE .OR. IP .LT. 2 .OR.
C      *   N .LE. IP) RETURN
C      IFAULT = 3
C      IF (CDF .LT. ZERO .OR. CDF .GT. ONE) RETURN
C      IFAULT = 0
C      IF (CDF .EQ. ZERO .OR. CDF .EQ. ONE) RETURN
C
C      Use ONE and ZERO as two starting points for the
C      Illinois method
C
C      X0 = ONE
C      F0 = ONE - CDF
C      X1 = ZERO
C      F1 = -CDF
C
C      Continue iterations until convergence is achieved
C
C
10  DIFF = F1 * (X1 - X0) / (F1 - F0)
    X2 = X1 - DIFF
    F2 = SQMCOR(X2, IP, N, RHO2, IFAULT) - CDF
    IF (IFAULT .NE. 0) RETURN

```

```

C
C      Check for convergence
C
IF (ABS(F2) .LE. EPS * CDF) GO TO 20
IF (ABS(DIFF) .LE. EPS * X2) GO TO 20
IF (F2 * F1 .GE. ZERO) THEN
  F0 = F0 / TWO
ELSE
  X0 = X1
  F0 = F1
END IF
X1 = X2
F1 = F2
GO TO 10
C
20 SQMCQ = X2
RETURN
END

```

Algorithm AS 262

A Two-sample Test for Incomplete Multivariate Data

By Robert W. Makuch† and Michael Escobar

Yale University, New Haven, USA

and Samuel Merrill III

Wilkes College, Wilkes-Barre, USA

[Received July 1987. Final revision November 1989]

Keywords: Censored data; Gehan test; Incomplete data; Log-rank test; Multivariate test

Language

Fortran 77

Description and Purpose

Wei and Lachin (1984) described a family of asymptotically distribution-free tests for equality of two multivariate distributions based on censored data. These tests are natural generalizations of the log-rank test (Mantel, 1966) and the generalized Wilcoxon test of Gehan (1965), both used extensively in the comparison of time-to-failure distributions between two groups. These methods properly take into account the possibly censored nature of events that contain only partial information about the random variables of interest. The generalized tests are obtained on the basis of the commonly used random censorship model (Kalbfleisch and Prentice, 1980), where the censoring vectors for each subject are mutually independent and also are independent of the underlying failure time vectors.

Censored multivariate time-to-event data are encountered frequently in clinical

†*Address for correspondence:* Department of Epidemiology and Public Health, Yale University School of Medicine, 60 College Street, New Haven, CT 06510, USA.